SIF5003 for F1 høsten 2002

Maple-øving 1: Løsningsforslag

M1-1: Sum, Riemannsum.

Vi skriver opp summen med Sum (som ikke prøver å regne ut summen), finner verdien med value (som gjør om Sum til sum, som regner ut summen om den klarer det), og plukker ut høyresiden av ligningen med rhs for senere bruk:

```
> Sum(i^4,i=1..n): %=value(%); s:=rhs(%):
```

$$\sum_{i=1}^{n} i^4 = \frac{1}{5}(n+1)^5 - \frac{1}{2}(n+1)^4 + \frac{1}{3}(n+1)^3 - \frac{1}{30}n - \frac{1}{30}$$

Funksjonen factor faktoriserer polynomer:

```
> factor(s);
```

$$\frac{1}{30} n (2n+1)(n+1)(3n^2+3n-1)$$

Suksess. Vi lagrer resultatet i `s` for senere bruk.

```
> s:=%:
>
```

Induksjonsbevis

Først basistrinnet:

```
> subs(n=1,s);
```

$$1$$

Så induksjonstrinnet. Vi tar uttrykket (`s`) vi skal vise er summen, legger til $(n+1)$-te ledd og trekker fra resultatet av å substituere $n+1$ for n i `s`:

```
> s+(n+1)^4-subs(n=n+1,s);
```

$$\frac{1}{30} n (2n+1)(n+1)(3n^2+3n-1) + (n+1)^4 - \frac{1}{30}(n+1)(2n+3)(n+2)(3(n+1)^2+3n+2)$$

Dette burde ha blitt null. I så fall skulle induksjonsbeviset være komplett. Hva skjer om vi forenkler?

```
> simplify(%);
```

$$0$$

```
>
```

Riemannsum

Riemannsum for $\int_0^1 x^4\, dx$ med intervallet $[0, 1]$ delt i n deler og bruk av høyre endepunkt i hvert delintervall:

```
> Sum((i*b/n)^4,i=1..n)*b/n: %=value(%);
```

$$\frac{\left(\sum_{i=1}^{n}\frac{i^4 b^4}{n^4}\right)b}{n}=\frac{\left(\frac{1}{5}\frac{b^4(n+1)^5}{n^4}-\frac{1}{2}\frac{(n+1)^4 b^4}{n^4}+\frac{\frac{1}{3}b^4(n+1)^3}{n^4}-\frac{1}{30}\frac{b^4(n+1)}{n^4}\right)b}{n}$$

```
> simplify(rhs(%));
```

$$\frac{1}{30}\frac{b^5(n+1)(6n^3+9n^2+n-1)}{n^4}$$

```
> Limit(%,n=infinity): %=value(%);
```

$$\lim_{n\to\infty}\frac{1}{30}\frac{b^5(n+1)(6n^3+9n^2+n-1)}{n^4}=\frac{1}{5}b^5$$

Og det er som forventet.

```
>
```

M1-2: Fjerdederivert, estimat for Simpson

Den fjerdederiverte av $\frac{\sin(x)}{x}$:

```
> diff(sin(x)/x,x,x,x,x); d4:=%:
```

$$\frac{\sin(x)}{x}+\frac{4\cos(x)}{x^2}-\frac{12\sin(x)}{x^3}-\frac{24\cos(x)}{x^4}+\frac{24\sin(x)}{x^5}$$

```
> plot(d4,x=0..4*Pi);
```



Åpenbart opptrer største tallverdi for $x=0$. Vi kan regne ut verdien der eksakt:

```
> limit(d4,x=0);
```

$$\frac{1}{5}$$

Estimat for Simpson på [0, π]:

```
> Sn_est:=(1/5)*Pi^5/(180*n^4);
```

$$Sn\_est := \frac{1}{900}\frac{\pi^5}{n^4}$$

Vi ønsker feil mindre enn $10^{(-6)}$:

```
> Sn_est=10^(-6); solve(%,n);
```

$$\frac{1}{900}\frac{\pi^5}{n^4} = \frac{1}{1000000}$$

$$\frac{10}{3}\sqrt{3}\,\pi^{(5/4)}, \frac{10}{3}I\sqrt{3}\,\pi^{(5/4)}, -\frac{10}{3}\sqrt{3}\,\pi^{(5/4)}, \frac{-10}{3}I\sqrt{3}\,\pi^{(5/4)}$$

Huffda, to reelle og to imaginære løsninger. Den vi er interessert i (den positive reelle roten) kc først:

```
> evalf(%[1]);
```

$$24.14775237$$

Så vi må bruke n=26 for å være helt sikre etter dette estimatet.

```
>
```

Utfordring (ikke obligatorisk): Regn ut S_n og sammenlign med integralet

```
> Simpson:=(f,a,b,n)->((b-a)/(3*n))*
    (f(a)+Sum((3-(-1)^i)*f(a+i*(b-a)/n),i=1..n-1)+f(b)):
> Simpson(f,a,b,n);
```

$$\frac{1}{3}\,\frac{(b-a)\left(f(a)+\left(\sum_{i=1}^{n-1}(3-(-1)^i)\,f\left(a+\frac{i\,(b-a)}{n}\right)\right)+f(b)\right)}{n}$$

```
> value(subs(n=8,a=0,b=8,%));
```

$$\frac{1}{3}f(0)+\frac{4}{3}f(1)+\frac{2}{3}f(2)+\frac{4}{3}f(3)+\frac{2}{3}f(4)+\frac{4}{3}f(5)+\frac{2}{3}f(6)+\frac{4}{3}f(7)+\frac{1}{3}f(8)$$

```
> Simpson(x->sin(x)/x,0.000001,Pi,26);
```

$$\frac{1}{78}(\pi-.1\,10^{-5})\left(1.000000000+\left(\sum_{i=1}^{25}\frac{(3-(-1)^i)\sin\left(.1\,10^{-5}+\frac{1}{26}i\,(\pi-.1\,10^{-5})\right)}{.1\,10^{-5}+\frac{1}{26}i\,(\pi-.1\,10^{-5})}\right)\right)$$

Her kommer endelig resultatet av Simpson for $n = 26$:

```
> evalf(%);
```

$$1.851936200$$

Her er integralet. Artig nok har Maple en navngitt funksjon Si for dette (integralsinus):

```
> Int(sin(x)/x,x=0..Pi): %=value(%);
```

$$\int_0^{\pi}\frac{\sin(x)}{x}\,dx = \mathrm{Si}(\pi)$$

```
> evalf(rhs(%));
```

$$1.851937052$$

```
>
```

M1-3: Rotasjonsflater og -legemer

Kurven / området vi skal rotere:

```
> plot(exp(-x^2),x=0..2,filled=true);
```



Definisjoner, kopiert fra rotasjonsflater.mws:

```
> rotxplot:=proc(expr,range)
    return
  plottools[transform]((x,theta,y)->[x,y*cos(theta),y*sin(th
  eta)])

  (plot3d(expr,range,theta=0..2*Pi,op(3..nargs,[args])));
  end:
  rotyplot:=proc(expr,range)
    return
  plottools[transform]((x,theta,y)->[x*cos(theta),x*sin(thet
  a),y])

  (plot3d(expr,range,theta=0..2*Pi,op(3..nargs,[args])));
  end:
  rotxploty:=proc(expr,range)
    return
  plottools[transform]((y,theta,x)->[x,y*cos(theta),y*sin(th
  eta)])

  (plot3d(expr,range,theta=0..2*Pi,op(3..nargs,[args])));
  end:
  rotyploty:=proc(expr,range)
    return
  plottools[transform]((y,theta,x)->[x*cos(theta),x*sin(thet
```

```
a),y])

(plot3d(expr,range,theta=0..2*Pi,op(3..nargs,[args])));
end:
```

Rotert om x-aksen

```
> rotxplot(exp(-x^2),x=0..2,scaling=constrained,axes=boxed);
```




```
> Int(Pi*y^2,x=0..2);
```

$$\int_0^2 \pi\, y^2\, dx$$

```
> subs(y=exp(-x^2),%);
```

$$\int_0^2 \pi\, (\mathbf{e}^{(-x^2)})^2\, dx$$

```
> value(%);
```

$$\frac{1}{4}\,\mathrm{erf}(2\,\sqrt{2}\,)\,\pi^{(3/2)}\,\sqrt{2}$$

```
> evalf(%);
```

$$1.968576540$$

```
>
```

Rotert om y-aksen

```
> rotyplot(exp(-x^2),x=0..2,scaling=constrained,axes=boxed);
```




```
> Int(2*Pi*x*y,x=0..2);
```

$$\int_0^2 2\,\pi\,x\,y\,dx$$

```
> subs(y=exp(-x^2),%);
```

$$\int_0^2 2\,\pi\,x\,\mathbf{e}^{(-x^2)}\,dx$$

```
> value(%);
```

$$-\pi\,\mathbf{e}^{(-4)}+\pi$$

```
> evalf(%);
```

$$3.084052377$$

```
>
```